ABSTRACT
This Hands-on-Workshop will introduce many of the Actions, Options and Sub-Options of the New ODS EXCEL Destination, and give the students experience visualizing the results of the new ODS EXCEL Destination outputs by executing code and viewing the results. This class will cover the some of the topics in the following subject areas:

- Differences between SAS Tagsets and SAS ODS Destinations
- SAS ODS EXCEL Destination Actions
- How to Set Excel Property Values
- Modifying Output Features Using Anchor, Style, and CSS
- Options that Apply to Titles, Footnotes, and Sheet processing
- Ways you can adjust Print Features like Headers and Margins
- Options that Impact Columns, Rows, and Cells

INTRODUCTION
Writing data from SAS data sets to Excel workbooks has long been a requested task of SAS Programmers. Base SAS has had many ways to perform this task. You could “Right-Click” on a SAS data set icon in the SAS display manager window, use the SAS PROC EXPORT, a LIBNAME statement, SAS Tagsets like CSV and HTML could create Excel readable text files. Other Tagsets like EXCELXP had the capability to format the output data and worksheets, with many options. Some methods like HTML could include graphs, if you could figure out how to code for writing them. The EXCELXP tagset is a good, flexible, and reliable tool for outputting data from SAS data sets to Excel workbooks. Some of the methods were restricted to writing only one worksheet within an Excel workbook while other methods can write several. When SAS Institute released Base SAS Version 9.4 TS Level 1M1 [1, page 147] in 2015 they sent the ODS Excel Destination into user hands as an experimental feature to upgrade how SAS programmers were able to move data from SAS to Excel. The first supported version is 9.4 TS Level 1M3.

This Hands-On-Workshop (HOW) will focus on the features of the ODS EXCEL Destination. I will spend time describing the differences between the SAS Tagsets and SAS ODS Destinations. Then as the name implies, I will move on to examples that you as a student can get your hands onto to see how the system treats both data and output activities.

Types of ODS EXCEL Destination Actions and Options
While the EXCELXP tagset and the ODS EXCEL Destination are separate and unique features of SAS, many of the options of each tool, provide similar output to an output Excel workbook. I will describe some of the arguments and options while providing as many examples as space and time permits. The goal is to show something for each group. The lists below show my general grouping categories. I will describe something from each of these general groupings [2, page 3] and options later. In an 80-minute class, I may need to work with several of these features in one example.
The ODS Excel Destination: Assorted Tips and Techniques to Simplify Writing SAS® Data to Excel Workbooks, cont.

ODS Destination EXCEL Arguments
- File identification
- Excel file properties
- Output features

ODS Destination EXCEL Option Groups
- Workbook
- Worksheet
- Print
- Column
- Row
- Cell level

Most of the time when a software manufacturer produces a manual describing their software the commands are in alphabetical order. But I do not write code in alphabetical order. If I want to modify a row of output I look for commands that modify rows. Indices should be in alphabetical order, not manuals. This paper will be grouping the projects in such a way as to perform groups of tasks that impact similar areas of Excel workbooks and spreadsheets.

A SAS ODS Tagset VS an ODS Destination
OK, it is true that both Tagsets and Destinations are part of the SAS Output Delivery System (ODS for short). The main difference is that Tagsets are PROC TEMPLATE code modules stored in the depths of the SAS Software Environment, and are compiled and executed at run-time. If you can find them you can change them [1, page 110], if you dare. SAS is usually nice in that their tagsets are protected, but you can create one with the same name and put it into the same path as the SAS version, making your tagset override their tagset. You can even create your own tagsets. Destinations however are “Black-Box” entities that stand alone and cannot be modified by the common programmer. I will not provide any code or workshop examples of the tagsets, because we want to see how the ODS EXCEL Destination works. See Chapter 8 of reference [1] for more details about Tagsets.

Examples of the ODS EXCEL Destination Arguments
The first project is an example of the least amount of SAS code needed to create an output Excel workbook, this project uses only default conditions to do the work. But first let’s look at the general syntax of the ODS EXCEL Destination code. The general syntax of the ODS EXCEL statement is rather simple, but the options are many and varied. Note that this book discusses “Actions”, “Options”, and “Sub-Options.” For the most complete and updated list of ODS EXCEL options, refer to the ODS documentation at the following URL: http://support.sas.com/documentation/cdl/en/odsug/69832/PDF/default/odsug.pdf. See Chapter 6 for detailed information and examples of the ODS Excel destination.

SAS Code 1 – Simple ODS Excel Syntax to open and close an Output path to an Excel workbook.

```sas
* Syntax to open and close the ODS EXCEL destination. ;
ODS EXCEL <(<ID=> identifier)> < action> ; * Open form 1;
ODS EXCEL <(<ID=> identifier)> <option(s)> ; * Open form 2;
ODS EXCEL <(<ID=> identifier)> CLOSE; * Close ;
```

The SAS ODS Excel destination syntax shown in SAS Code 1 is just the tip of the iceberg. As shown, everything except “ODS EXCEL,” or “CLOSE” is optional.

File identification
This section will show the default execution and location of an Excel workbook, how to name an Excel workbook, and writing more than one Workbook with different output options. It includes Project 0 to Project 3.

- Project 0 – Build a Path macro variable
- Project 1 – Use default execution and locate of an Excel workbook
- Project 2 - Name an Excel workbook
- Project 3 - Writing Multiple Workbooks with different output options at the same time
Project 0 – Build a Path macro variable
First the run SAS code to build a SAS Macro variables and clean the work areas for this workshop.

Project 0 – SAS Code to create a SAS Macro Variable Path Name.

```sas
/*----------------------------------------------------------
** This code defines the path macro variable and sets **
** up the system for the class. It clears the output  **
** directories and SAS work space, then builds         **
** FILENAME references to the exercises.              **
** The %LET Path varaible needs to be executed on each **
** computer in the HOW room.                          **
**----------------------------------------------------------*/
%global path;
*let path = C:\HOW\Benjamin;
*let path = D:\sasdata\HOW\Benjamin;
%let path = F:\2018 Conferences\WUSS 2018\my_papers\Paper 153 - HOW\HOW\Benjamin;
*let path = F:\2018 Conferences\SESUG 2018\Paper 165\HOW\Benjamin;
%macro set_up;
%put _user_;
options mprint mlogic symbolgen mprintnest mlogicnest;
* remove output files if they exist;
* start fresh each time;
options xsync noxwait;
x "del &path.\out\*./* /Q";
* NOTE a possible warning may appear if no "WORK" files exist;
proc datasets library=work kill memtype=data details;
run;
quit;
filename solution "&path./solutions";
filename exercise "&path./exercises";
```
%mend set_up;
%set_up;
%let file1  = &path\out\excel_Default_STYLE.xlsx;
%let file2  = &path\out\excel_Harvest_STYLE.xlsx;
%let file3  = &path\out\excel_Moonflower_STYLE.xlsx;
%let file4  = &path\out\excel_Property_changes.xlsx;
%let file5a = &path\out\project_5_a.xlsx;
%let file5b = &path\out\project_5_b.xlsx;
%let file6a = &path\out\project_6_a.xlsx;
%let file6b = &path\out\project_6_b.xlsx;
%let file7a = &path\out\project_7_a.xlsx;
%let file7b = &path\out\project_7_b.xlsx;
%let file8a = &path\out\project_8a_PRINT_HEADER_Margin_0.5.xlsx;
%let file8b = &path\out\project_8a_PRINT_HEADER_Margin_1.xlsx;
%let file8c = &path\out\project_8a_PRINT_HEADER_Margin_4.xlsx;
%let file9  = &path\out\project_9.xlsx;
%let file10 = &path\out\project_10.xlsx;

data ASIA_ONLY;
  set sashelp.shoes (where=(region="Asia"));
run;

Project 1 – The Simple Default No Option Code
First the SAS code to do this work it will be labeled as SAS Code 1 in the workshop.

Project 1 – SAS Code to create an Excel Workbook.

ODS EXCEL;
PROC PRINT DATA=sashelp.shoes;
RUN;
ODS EXCEL CLOSE;

The following is the SAS Results page after running the SAS Code in Project 1.
Project 1 – SAS Results Window.

This shows all columns and the first 13 rows of the SAS Results window. When you run the code for Project 1 you should get similar results. Notice the information in the lower left hand corner of the image. The path “C:\Users\wmebe_000” is shown at that location. The information on your computer is likely to be different than the information shown here. This folder is where you will find your output Excel Workbook.

SAS Project 1 Excel output workbook named “sasexcl.xlsx”.
This screen print is the output of the ODS EXCEL Destination code listed in the code box "Project 1 SAS Code to create an Excel Workbook." The Excel ribbon is collapsed to show more of the data generated.

**Project 2 – Name an Excel workbook**

**Project 2 – SAS Code to create an Excel Workbook.**

```
ODS EXCEL FILE="&path\out\My_Project_2_Workbook.xlsx";
PROC PRINT DATA=sashelp.shoes;
RUN;
ODS EXCEL CLOSE;
```

**Project 2 – SAS Results Viewer After Running Code to Create and Name an Excel Workbook.**

```
<table>
<thead>
<tr>
<th>Obs</th>
<th>Region</th>
<th>Product</th>
<th>Subsidiary</th>
<th>Stores</th>
<th>Sales</th>
<th>Inventory</th>
<th>Returns</th>
</tr>
</thead>
<tbody>
<tr>
<td></td>
<td>Africa</td>
<td>Boot</td>
<td>Addis Ababa</td>
<td>12</td>
<td>29,761</td>
<td>191,021</td>
<td>769</td>
</tr>
<tr>
<td></td>
<td>Africa</td>
<td>Men's Casual</td>
<td>Addis Ababa</td>
<td>4</td>
<td>67,242</td>
<td>118,036</td>
<td>2,284</td>
</tr>
<tr>
<td></td>
<td>Africa</td>
<td>Men's Dress</td>
<td>Addis Ababa</td>
<td>7</td>
<td>76,793</td>
<td>136,273</td>
<td>2,433</td>
</tr>
<tr>
<td></td>
<td>Africa</td>
<td>Sandal</td>
<td>Addis Ababa</td>
<td>10</td>
<td>62,819</td>
<td>204,284</td>
<td>1,961</td>
</tr>
<tr>
<td></td>
<td>Africa</td>
<td>Slipper</td>
<td>Addis Ababa</td>
<td>14</td>
<td>68,641</td>
<td>279,795</td>
<td>1,771</td>
</tr>
<tr>
<td></td>
<td>Africa</td>
<td>Sport Shoe</td>
<td>Addis Ababa</td>
<td>4</td>
<td>1,690</td>
<td>16,634</td>
<td>79</td>
</tr>
<tr>
<td></td>
<td>Africa</td>
<td>Women's Casual</td>
<td>Addis Ababa</td>
<td>2</td>
<td>51,541</td>
<td>98,841</td>
<td>940</td>
</tr>
<tr>
<td></td>
<td>Africa</td>
<td>Women's Dress</td>
<td>Addis Ababa</td>
<td>12</td>
<td>108,942</td>
<td>311,017</td>
<td>3,233</td>
</tr>
<tr>
<td></td>
<td>Africa</td>
<td>Boot</td>
<td>Algiers</td>
<td>21</td>
<td>21,297</td>
<td>73,737</td>
<td>770</td>
</tr>
<tr>
<td></td>
<td>Africa</td>
<td>Men's Casual</td>
<td>Algiers</td>
<td>4</td>
<td>63,296</td>
<td>100,982</td>
<td>2,221</td>
</tr>
<tr>
<td></td>
<td>Africa</td>
<td>Men's Dress</td>
<td>Algiers</td>
<td>13</td>
<td>133,743</td>
<td>428,575</td>
<td>3,621</td>
</tr>
<tr>
<td></td>
<td>Africa</td>
<td>Sandal</td>
<td>Algiers</td>
<td>25</td>
<td>28,198</td>
<td>84,447</td>
<td>1,530</td>
</tr>
<tr>
<td></td>
<td>Africa</td>
<td>Slipper</td>
<td>Algiers</td>
<td>17</td>
<td>64,891</td>
<td>240,198</td>
<td>1,823</td>
</tr>
<tr>
<td></td>
<td>Africa</td>
<td>Sport Shoe</td>
<td>Algiers</td>
<td>9</td>
<td>2,617</td>
<td>9,372</td>
<td>168</td>
</tr>
</tbody>
</table>
```

**Project 2 – The Named Excel Workbook**
Yes I know all of the images look just about the same, but that will change soon. See the new name on the top of the Excel worksheet. The next image is a screen print of where the file went.

Project 2 – Directory Screen Print showing location of the EXCEL output workbook.

Note the output file is labeled with an Excel *.xlsx file extension. If we rename and open the file with as a ZIP file we can see it really is in an *.xlsx file format.
Project 3 - Writing Multiple Workbooks with different output options at the same time

This project we will get a little more adventurous, and make the Excel output look a little bit different. You may not know this but every ODS output has a default “STYLE” associated with the output. For the ODS Excel Destination the default STYLE is EXCEL [2, page 63]. Project 3 will show some different styles for the fun of it.

![Excel Workbook](image)

**Project 3 – Writing multiple Workbooks in one SAS step.**

```
ODS EXCEL (ID=Default   ) File="&file1" ; * default style   ;
ODS EXCEL (ID=Harvest   ) File="&file2" STYLE=Harvest   ; * Harvest style   ;
ODS EXCEL (ID=Moonflower) File="&file3" STYLE=Moonflower; * Moonflower style;
Proc Print data=Asia_only;
run;

ODS EXCEL (ID=Default   ) CLOSE;       * Close Default   ;
ODS EXCEL (ID=Harvest   ) CLOSE;       * Close Harvest   ;
ODS EXCEL (ID=Moonflower) CLOSE;       * Close Moonflower;
```

**Table 1 – List of Supported SAS Styles for SAS 9.4 1M3**

<table>
<thead>
<tr>
<th>List of SAS Styles Supported (SAS 9.4 1M3)</th>
</tr>
</thead>
<tbody>
<tr>
<td>Analysis</td>
</tr>
<tr>
<td>Daisy</td>
</tr>
<tr>
<td>Excel</td>
</tr>
<tr>
<td>Gantt</td>
</tr>
<tr>
<td>HighContrast</td>
</tr>
<tr>
<td>Journal2</td>
</tr>
<tr>
<td>Listing</td>
</tr>
<tr>
<td>MonochromePrinter</td>
</tr>
<tr>
<td>NoFontDefault</td>
</tr>
<tr>
<td>Pearl</td>
</tr>
<tr>
<td>PowerPointLight</td>
</tr>
<tr>
<td>Sapphire</td>
</tr>
<tr>
<td>SeasidePrinter</td>
</tr>
<tr>
<td>vaDark</td>
</tr>
</tbody>
</table>
This list can be produced with the following SAS Code.

```sas
ods _all_ close;
ods listing;
proc template;
  list styles;
run;
quit;
```

Now let's look at the Excel output files.

**Project 3 – Default style**

![Excel spreadsheet example](image)

**Project 3 – Harvest style**

![Excel spreadsheet example](image)
Project 3 – Moonflower style

Excel file properties

We have already seen the FILE property that we can add to an output Excel workbook, yes giving the output file a name is adding a property to the file. There are several other properties [2, page 23] but here we will only look at four more. I have chosen “AUTHOR”, “CATEGORY”, “COMMENTS”, and “KEYWORDS”. We can do this in one step before the file is opened by anyone, this could really save some time if you needed to add comments to 100 workbooks.

Project 4 – Adding properties “AUTHOR”, “CATEGORY”, “COMMENTS”, and “KEYWORDS” to a Workbook

Project 4 – SAS code to modify some properties of the Output Excel workbook.

```
ODS EXCEL FILE="&file4"
AUTHOR = 'AUTHOR - Put your name here'
CATEGORY = 'CATEGORY - dream up a new Category name'
COMMENTS = 'COMMENTS - Your chance to make a comment'
KEYWORDS = 'KEYWORDS - Nice to be able to add keywords and not change the file'
;
PROC PRINT DATA=Asia_only;
RUN;
ODS EXCEL CLOSE;
```
Project 4 – Excel Property sheet Updated by SAS at run-time.

Output features
The output features I will discuss here are the use of Anchors, Cascading Style Sheets (CSS), and fuzziness of graphs that can impact the data and images placed into your Excel workbooks. Anchors and style sheets work together to manipulate how data is displayed in the output process. The default anchor name with SAS processed style sheets is "idx". Here in Project 5, I will use the anchor name I made up called “Expense” to replace ‘idx’. Then I will make changes to two “Classes” of information in the style sheet. One called “Header” and one called “data”. The changes I will make are to the “background-color” of each of these classes of data. Project 6 will address the fuzziness of the output by building a graph and displaying the output with different Dot-Per-Inch (DPI) densities. Once again, this is a simple starter graph using SAS data available to everyone. I am sure you can make a better graph.

Project 5- What are Anchors and Cascading Style Sheets?
To the novice user of Cascading Style Sheets or SAS Style sheets anchors are the mystery of ODS. I spent weeks researching ODS ANCHORS and have a better definition elsewhere [2, chapter 6]. But time and space here does not permit going any deeper than to say that the ODS EXCEL DOM <"external-file"> command will expose them. For now all we need to know is that they exist and we can create a simple CSS text file to use them with the SAS ODS EXCEL Destination commands. We will use the code in Project 5 to create and use a simple CSS text file to use anchor commands to modify the output of an EXCEL workbook using an ODS ANCHOR.

An Anchor tag is an important part of the ODS output. You can specify an “ANCHOR” or allow ODS to generate them using default names. The syntax is to provide the keyword ANCHOR and an anchor-name, as follows: ANCHOR=“anchor-name". The default anchor name is "idx". Once you specify the first anchor, that name is used and a number is added to the end of the name for the next anchor.

The code in “Project 5 – Build a CSS Text…” creates a user generated Cascading Style Sheet with two references called “#Expense .header” and ‘#Expense .data”. Each of these references modify some portion of the Excel output.
With “#Expense .header” I am changing the background color of the headers in the Excel worksheet and with “#Expense .data” I am changing the background colors of the data lines.

When you use an ANCHOR, you are changing the ID= attribute of the current style sheet output in use, this is effective in both SAS style sheets and Cascading style sheets (#ID=). To identify where these anchors appear in the output style, use the ODS DOM option. There are many documents, papers, and books written about CSS Style sheets in which you can find more examples. The official website of the World Wide Web Consortium that maintains the CSS standards is https://www.w3.org/Style/CSS/. Another good source of information about this type of data manipulation is a SAS Global Forum paper found at the following URL: http://support.sas.com/resources/papers/proceedings15/SAS1880-2015.pdf

Project 5 – Build a CSS Text File and Write Two EXCEL Workbooks Using an ANCHOR and Not Using One.

```sas
* Build a Cascading Style Sheet file;
filename my_css "css_text.css";
data null;
  file my_css noprint notitles linesize=256;
  put
    '#Expense .header {background-color:orange}' /
    '#Expense .data   {background-color:pink }' ;
run;
ods html close;

* do not use a Cascading Style Sheet file;
ods excel (ID=File5a) File="&file5a" options(sheet_interval='none');

* Use a Cascading Style Sheet file with an anchor;
ods excel (ID=File5b) File="&file5b" cssstyle= "css_text.css" anchor="expense"
   options(sheet_interval='none');
proc print data=sashelp.shoes(where=(
  Product = 'Boot' and
  Subsidiary = 'New York' and
  region     = 'United States'));
run;
ods excel (ID=File5a) close;
ods excel (ID=File5b) close;
ods html;
```
Project 5 – Excel output without using an anchor and CSS test file.

![Excel output without anchor and CSS test file](image1)

Project 5 – Excel output using an anchor and CSS test file.

![Excel output with anchor and CSS test file](image2)

Project 6 – How the Dots Per Inch Effect Fuzziness of the Outputs.

Project 6 deals with showing how the Dots-Per-Inch (DPI) densities impact the output image generated when images are sent to Excel. The main issue occurs when digital images are expanded and how they look around the edges. The graph in Project – 6 is a simple graph of data available to all SAS Programmers. I also threw in some PROC SQL code to mix things up a little bit.

![Graph showing fuzziness](image3)
Project 6 – SAS Graph Macro and SAS Code to Create Two Excel Workbooks.

```sas
%macro Graph_it;
PROC SQL;
  CREATE VIEW WORK.Sorted_1 AS
    SELECT T.Region, T.Sales
    FROM SASHELP.SHOES(WHERE=(Region ne "Asia")) as T;
QUIT;
Legend1
  FRAME
  POSITION = (BOTTOM CENTER OUTSIDE);
TITLE1 "Pie Chart of SASHELP.Shoes Return data by Region";
TITLE2 "Except Asia";
PROC GCHART DATA =WORK.Sorted_1;
  PIE3D Region / SUMVAR = Sales
    TYPE = SUM LEGEND=LEGEND1
    SLICE = OUTSIDE
    PERCENT = OUTSIDE
    VALUE = OUTSIDE
    OTHER = 4
    OTHERLABEL = "Other"
    COUTLINE = BLACK
    NOHEADING;
RUN;
QUIT;
TITLE;
FOOTNOTE;
RUN;
%mend Graph_it;

ods html close;

*******************************************************************************;
* output graph using the default image_dpi of 300 **;
*******************************************************************************;
ods excel (ID=File6a) File="&file6a";

*******************************************************************************;
* output graph using the modified image_dpi of 700 **;
*******************************************************************************;
ods excel (ID=File6b) File="&file6b" image_dpi=700;
%graph_it;
ods excel (ID=File6a) close;
ods excel (ID=File6b) close;
ods html;
```
Project 6 – SAS Graph Output “A” in Excel workbook at 300 DPI.
Project 6 – SAS Graph Output “B” in Excel workbook at 700 DPI.

Project 7 - Options that Apply to Titles, Footnotes, and Sheet processing

What can you do with titles, and footnotes? Titles go on top of the page and footnotes go on the bottom of the page, right! Not always when you are using the ODS EXCEL Destination. When you consider worksheet processing options then titles and footnotes can be scattered all over the worksheets. Each title and footnote can be turned on, turned off, displayed once or many times. Even the output data does not need to stay the same place every time a SAS process is executed. The data can be split by “BYGROUP”, “PAGE”, “PROC”, “TABLE”, or not at all. Time permits only one example for this topic, so I choose to output two tables on one page with multiple titles and multiple footnotes. You can play with the other options when you get home [2, chapter 7].

The SAS Code in project 7 writes two Excel output Workbooks. The first workbook Project 7 “A” deals with titles and the PAGE_INTERVAL = 'none'. This example project turns on all titles and displays only titles and no footnotes. Notice that it shows Title “ONE” and Title “TWO” with table output data from two PROC PRINT procedures all on the same worksheet page. The default would have been to write two separate worksheets, each with a separate and different title on each worksheet, when EMBEDDED_TITLES=’on’ is active.

The second workbook Project 7 “B” deals with footnotes and the PAGE_INTERVAL = 'none'. This example project turns on only the last footnote and displays no titles. Notice that it shows only Footnote “TWO” with table output data from two PROC PRINT procedures all on the same worksheet page. The default would have been to write two separate worksheets, each with a separate and different footnote on each worksheet, when EMBEDDED_FOOTNOTES=’on’ is active.
Project 7 - SAS Code to Display Titles, Footnotes, and Worksheet Options.

```sas
Title 'title one';
Footnote 'foot note one';
ods excel (ID=File7a) File = "&file7a"
   options(EMBEDDED_TITLES = 'on'
            SHEET_INTERVAL = 'none');
ods excel (ID=File7b) File="&file7b"
   options(EMBEDDED_FOOTNOTES = 'on'
            EMBED_FOOTNOTES_ONCE = 'on'
            SHEET_INTERVAL       = 'none');
   proc print data=Asia_only;
   run;
Title 'title two';
Footnote 'foot note Two';
   proc print data=Asia_only;
   run;
ods excel (ID=File7a) close;
ods excel (ID=File7b) close;
```

Project 7 – Titles and sheet interval output, All titles.

![Excel outputs with titles and footnotes]
Project 7 – Footnote and sheet interval output, Last Footnote.

![Excel output screenshot]

Project 8 – Print Header and Footer Margin Options
In addition to SAS Titles and Footnotes you can manipulate Excel Headers and Footers, and how they are spaced on the printed output pages. The SAS Code shown in Project 8 – SAS code produces Excel workbooks with different spacing of the Excel Headers and Excel Footers. They are one half inch from the top and the bottom, one inch from the top and bottom, and 4 inches from the top and the bottom. I further restricted the printed output to six rows and five columns so I could show part of the EXCEL Header that would normally be hidden. To show the impact of this SAS Code I needed to display the print image from Excel Property sheets and turn on the margin indicators. Other features also exist [2, Chapter 8]

Project 8 – SAS Code to Show EXCEL Headers and Footers.

```sas
title;
ods excel file = "&File8a"
   options(PRINT_HEADER='My Header Text' PRINT_FOOTER='My Footer Text'
   PRINT_HEADER_MARGIN='0.5' PRINT_FOOTER_MARGIN='0.5'
   PRINT_AREA='b,4,f,9');
proc print data=Asia_only;
run;
ods excel close;
ods excel file = "&File8b"
   options(PRINT_HEADER='My Header Text' PRINT_FOOTER='My Footer Text'
   PRINT_HEADER_MARGIN='1' PRINT_FOOTER_MARGIN='1'
   PRINT_AREA='b,4,f,9');
proc print data=Asia_only;
run;
ods excel close;
ods excel file = "&File8c"
   options(PRINT_HEADER='My Header Text' PRINT_FOOTER='My Footer Text'
   PRINT_HEADER_MARGIN='4' PRINT_FOOTER_MARGIN='4'
```
PRINT_AREA='b,4,e,9');
proc print data=Asia_only;
run;
ods excel close;

Project 8 – One Half Inch Margins Top and Bottom.
Project 8 – One Inch Margins Top and Bottom.

Project 8 – Four Inch Margins Top and Bottom.

Project 9 - Options that Impact Columns, Rows, and Cells.
Manual reformatting of an Excel Spreadsheet is at best time consuming and at worst tedious and error prone. So why do it? The SAS ODS EXCEL Destination software can do most tasks for you. Here in Project 9 I will show you how to Adjust Column widths, Apply Excel “Auto-Filtering” to a column, Freeze the header rows, Hide a one or more columns, Adjust Row heights, and start outputting your output data somewhere other than column 1 row 1. And that is not all that you can do [2, Chapter9].

Project 9 – SAS Code to adjust Rows, Columns, and Cells of a spreadsheet.

```sas
title1;
footnote1;
ods excel file = "&file9"
options(Absolute_Column_Width= '4,4,10,15,15,6,12,11,11'
  AUTOFILTER = '2'
  Frozen_Headers = 'on'
  Hidden_Columns = '6'
  Absolute_Row_Height = '15,30'
  start_at = '4,5'
  );
proc print data= ASIA_ONLY noobs;
run;
ods excel close;
```

Project 9 – EXCEL Output from Project 9 SAS Code adjusting Rows, Columns, and Cells of a spreadsheet.

The column “F” is hidden, and columns “A” and “B” are narrow while others are different sizes. The rows alternate in height and the “Product” column is filtered. And the most fun part of all is the fact that the output data does not start in Column 1 Row 1.
Project 10 – the ODS EXCEL Destination ROW_HEIGHTS option.
The ‘HEIGHTS’ part of the suboption name ROW_HEIGHTS is plural, and as the plural name implies, the suboption has an effect on several row heights. A single value of zero indicates that the font size should be used as the height of all of the row types described in the list below. For this option you need to provide a quoted string of positional parameters to identify the seven row heights listed below. Values can be skipped but skipped parameters require a comma to show that the parameter is missing, except the last used value. These row heights are measured in points.

- Table header rows
- Table body rows
- BY value lines
- Titles
- Footers
- Page break height
- Paragraph skip height

Project 10 – SAS ODS EXCEL Destination Code to Demonstrate ROW_HEIGHTS Sub-option.

```sas
ods excel file = "&file10"
   options(ROW_HEIGHTS = '20,15,15,35,55,25,5'
            embedded_titles = 'on'
            embedded_footnotes = 'on'
            sheet_interval = 'none');

Title 'my title';
footnote 'my footnote';

proc print data=asia_only;
   by Subsidiary;
run;

ods excel close;
```

The Excel screen-print shown below shows a title, a footnote, and various spacing areas between the output elements of the Excel worksheet. The code turned on both embedded titles and embedded footnotes in addition to placing all of the BY-Grouped data onto one worksheet.
CONCLUSION

This paper and Hands-On-Workshop has been a quick review of some of the features of the SAS ODS EXCEL Destination. Many of the actions, options, and sub-options covered in this class have other parameters that can be applied which make the options perform differently. At one time I counted over 125 independent actions, options, and sub-options. This makes the total number of ways you can modify an Excel workbook a very large number. The important thing to remember is that all of these actions are done while the Excel Workbook is being generated. That means you do not have to open the workbook to modify pieces and parts to make it look pretty. I did not show it but you can include images for the background, and far more complicated CSS files.
REFERENCES


CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Name: William E Benjamin Jr  
Enterprise: Owl Computer Consultancy, LLC  
Phone: 623-337-0269  
E-mail: wmebenjaminjr3@juno.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.
Appendices

Project 0 SAS Code

/* actual path will be assigned in class - make one up if you are at home*/

/*%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%*/
/**                                                      **/
/**  How - The ODS Excel Destination: Assorted Tips and  **/
/**        Techniques to Simplify Writing SAS® Data to   **/
/**        Excel Workbooks                               **/
/**  BY    William E Benjamin Jr,                        **/
/**        Owl Computer Consultancy, LLC,                **/
/**        Phoenix AZ.                                   **/
/**%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%*/
/**  This code defines the path macro variable and sets  **/
/**  up the system for the class. It clears the output   **/
/**  directories and SAS work space, then builds         **/
/**  FILENAME references to the exercises.               **/
/**                                                      **/
/**  The %LET Path varaible needs to be executed on each **/
/**  computer in the HOW room.                           **/
/**                                                      **/
/**%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%*/

/*%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%*/
/* set this to the input path for HOW */
/* exercises and solution code */
/*%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%*/
%global path;
*let path = C:\HOW\Benjamin;
%let path = D:\sasdata\HOW\Benjamin;

%macro set_up;

%put _user_;

options mprint mlogic symbolgen mprinthead mlogicnest;

* remove output files if they exist;
* start fresh each time;
options xsync noxwait;
%let path = &path.
%put x %del \out\*.* /Q;
* NOTE a possible warning may appear if no "WORK" files exist;
proc datasets library=work kill memtype=data details;
run;
quit;

filename solution "&path./solutions";
filename exercise "&path./exercises";

%mend set_up;
%set_up;

%let file1 = &path\out\excel_Default_STYLE.xlsx;
%let file2 = &path\out\excel_Harvest_STYLE.xlsx;
%let file3 = &path\out\excel_Moonflower_STYLE.xlsx;
%let file4 = &path\out\excel_Property changes.xlsx;
%let file5a = &path\out\project_5_a.xlsx;
%let file5b = &path\out\project_5_b.xlsx;
%let file6a = &path\out\project_6_a.xlsx;
%let file6b = &path\out\project_6_b.xlsx;
%let file7a = &path\out\project_7_a.xlsx;
%let file7b = &path\out\project_7_b.xlsx;
%let file8a = &path\out\File8a_PRINT_HEADER_Margin_0_5.xlsx;
%let file8b = &path\out\File8a_PRINT_HEADER_Margin_1.xlsx;
%let file8c = &path\out\File8a_PRINT_HEADER_Margin_4.xlsx;
%let file9 = &path\out\project_9.xlsx;
%let file10 = &path\out\project_10.xlsx;

data ASIA_ONLY;
set sashelp.shoes (where=(region="Asia"));
run;

**Project 1 SAS Code**

ODS EXCEL;
PROC PRINT DATA=sashelp.shoes;
RUN;
ODS EXCEL CLOSE;

**Project 2 SAS Code**

ODS EXCEL FILE="&path\out\My_Project_2_Workbook.xlsx";
PROC PRINT DATA=sashelp.shoes;
RUN;
ODS EXCEL CLOSE;
Project 3 SAS Code

ODS EXCEL (ID=Default   ) File="&file1"                 ; * default style   ;
ODS EXCEL (ID=Harvest   ) File="&file2" STYLE=Harvest   ; * Harvest style   ;
ODS EXCEL (ID=Moonflower) File="&file3" STYLE=Moonflower; * Moonflower style;

Proc Print data=Asia_only;
run;

ODS EXCEL (ID=Default   ) CLOSE;       * Close Default   ;
ODS EXCEL (ID=Harvest   ) CLOSE;       * Close Harvest   ;
ODS EXCEL (ID=Moonflower) CLOSE;       * Close Moonflower;

Project 4 SAS Code

ODS EXCEL FILE="&file4"
AUTHOR   = 'AUTHOR - Put your name here'
CATEGORY = 'CATEGORY - dream up a new Category name'
COMMENTS = 'COMMENTS - Your chance to make a comment'
KEYWORDS = 'KEYWORDS - Nice to be able to add keywords and not change the file'
;
PROC PRINT DATA=Asia_only;
RUN;
ODS EXCEL CLOSE;

Project 5 SAS Code

* Build a Cascading Style Sheet file;
filename my_css "css_text.css";
data _null_
   file my_css noprint notitles linesize=256;
   put
      '#Expense .header {background-color:orange}' /
      '#Expense .data   {background-color:pink }';
run;
Ods html close;

* do not use a Cascading Style Sheet file;
ods excel (ID=File5a) File="&file5a" options(sheet_interval='none');

* Use a Cascading Style Sheet file with an anchor;
ods excel (ID=File5b) File="&file5b" cssstyle = "css_text.css" anchor="expense"
options(sheet_interval='none');

proc print data=sashelp.shoes(where=
  (Product = 'Boot' and
   Subsidiary = 'New York' and
   region = 'United States'));
run;

ods excel (ID=File5a) close;
ods excel (ID=File5b) close;
ods html;

Project 6 SAS Code

%macro Graph_it;
  PROC SQL;
    CREATE VIEW WORK.Sorted_1 AS
      SELECT T.Region, T.Sales
      FROM SASHELP.SHOES(WHERE=(Region ne "Asia")) as T;
  QUIT;
Legend1
    FRAME
    POSITION = (BOTTOM CENTER OUTSIDE);
TITLE1 'Pie Chart of SASHELP.Shoes Return data by Region';
TITLE2 'Except Asia';

PROC GCHART DATA =WORK.Sorted_1;
  PIE3D Region / SUMVAR = Sales
    TYPE = SUM LEGEND=LEGEND1
    SLICE = OUTSIDE
    PERCENT = OUTSIDE
    VALUE = OUTSIDE
    OTHER = 4
    OTHERLABEL = "Other"
    COUTLINE = BLACK
    NOHEADING;
  RUN;
  QUIT;

TITLE;
FOOTNOTE;
RUN;
%mend Graph_it;
ods html close;

*****************************************************;
* output graph using the default image_dpi of 300 **;
******************************************************************************;
ods excel (ID=File6a) File="&file6a";

******************************************************************************;
* output graph using the modified image_dpi of 700 **;
******************************************************************************;

ods excel (ID=File6b) File="&file6b" image_dpi=700;
%graph_it;

ods excel (ID=File6a) close;
ods excel (ID=File6b) close;
ods html;

Project 7 SAS Code

Title 'title one';
Footnote 'foot note one';

ods excel (ID=File7a) File="&file7a"
opts(EMBEDDED_TITLES='on'
SHEET_INTERVAL='none');

ods excel (ID=File7b) File="&file7b"
opts(EMBEDDED_FOOTNOTES='on'
EMBED_FOOTNOTES_ONCE='on'
SHEET_INTERVAL='none');

proc print data=Asia_only;
run;

Title 'title two';
Footnote 'foot note Two';

proc print data=Asia_only;
run;
ods excel (ID=File7a) close;
ods excel (ID=File7b) close;
Project 8 SAS Code

title ;
ods excel file = "&File8a"
options(PRINT_HEADER='My Header Text' PRINT_FOOTER='My Footer Text'
PRINT_HEADER_MARGIN='0.5' PRINT_FOOTER_MARGIN='0.5'
PRINT_AREA='b,4,f,9');
proc print data=Asia_only;
run;
ods excel close;

ods excel file = "&File8b"
options(PRINT_HEADER='My Header Text' PRINT_FOOTER='My Footer Text'
PRINT_HEADER_MARGIN='1' PRINT_FOOTER_MARGIN='1'
PRINT_AREA='b,4,f,9');
proc print data=Asia_only;
run;
ods excel close;

ods excel file = "&File8c"
options(PRINT_HEADER='My Header Text' PRINT_FOOTER='My Footer Text'
PRINT_HEADER_MARGIN='4' PRINT_FOOTER_MARGIN='4'
PRINT_AREA='b,4,f,9');
proc print data=Asia_only;
run;
ods excel close;

Project 9 SAS Code

title1;
footnote1;

ods excel file = "&file9"
options(ABSOLUTE_COLUMN_WIDTH = '4,4,10,15,15,15,6,12,11,11'
AUTOFILTER = '2'
Frozen_Headers = 'on'
Hidden_Columns = '6'
ABSOLUTE_ROW_HEIGHT = '15,30'
start_at = '4,5'
);

proc print data= ASIA_ONLY noobs;
run;
ods excel close;

Project 10 SAS Code
ods excel file = "&file10"
  options(ROW_HEIGHTS = '20,15,15,35,55,25,5'
         embedded_titles = 'on'
         embedded_footnotes = 'on'
         sheet_interval = 'none');

Title 'my title';
footnote 'my footnote';

proc print data=asia_only;
  by Subsidiary;
run;

ods excel close;